F07ASFP (PZGETRS)

NAG Parallel Library Routine Document

Note: Before using this routine, please read the Users' Note for your implementation to check for implementation-dependent details. You are advised to enclose any calls to NAG Parallel Library routines between calls to Z01AAFP and Z01ABFP.

1 Description

F07ASFP (PZGETRS) solves an n by n complex system of linear equations with multiple right-hand sides, i.e., $A_s X = B_s$ or $A_s^T X = B_s$ or $A_S^H X = B_s$, where A_s is a submatrix of a larger m_A by n_A matrix A, i.e.,

$$A_s(1:n, 1:n) \equiv A(i_A : i_A + n - 1, j_A : j_A + n - 1),$$

and B_s is a (r right-hand sides) submatrix of a larger m_B by n_B matrix B, i.e.,

$$B_s(1:n, 1:r) \equiv B(i_B : i_B + n - 1, j_B : j_B + r - 1).$$

The matrix A_s must have been previously factorized by a call to F07ARFP (PZGETRF). F07ARFP (PZGETRF) performs an LU factorization and F07ASFP (PZGETRS) solves the system of equations by forward and backward substitution.

2 Specification

```
      SUBROUTINE F07ASFP(TRANS, N, NRHS, A, IA, JA, IDESCA, IPIV, B, IB,
     1                   JB, IDESCB, INFO)
      ENTRY      PZGETRS(TRANS, N, NRHS, A, IA, JA, IDESCA, IPIV, B, IB,
     1                   JB, IDESCB, INFO)
      COMPLEX*16         A(*), B(*)
      INTEGER            N, NRHS, IA, JA, IDESCA(9), IPIV(*), IB,
     1                   JB, IDESCB(9), INFO
      CHARACTER*1        TRANS
```

The ENTRY statement enables the routine to be called by its ScaLAPACK name.

3 Data Distribution

3.1 Definitions

The following definitions are used in describing the data distribution within this document:

m_p	–	the number of rows in the logical processor grid.
n_p	–	the number of columns in the logical processor grid.
p_r	–	the row grid coordinate of the calling processor.
p_c	–	the column grid coordinate of the calling processor.
M_b^X	–	the blocking factor for the distribution of the rows of a matrix X.
N_b^X	–	the blocking factor for the distribution of the columns of a matrix X.
numroc$(\alpha, b_\ell, q, s, k)$	–	a function which gives the **num**ber of **r**ows **o**r **c**olumns of a distributed matrix owned by the processor with the row or column coordinate q (p_r or p_c), where α is the total number of rows or columns of the matrix, b_ℓ is the blocking factor used (M_b^X or N_b^X), s is the row or column coordinate of the processor that possesses the first row or column of the distributed matrix and k is either n_p or m_p. The Library provides the function Z01CAFP (NUMROC) for the evaluation of this function.

3.2 Global and Local Arguments

The input arguments TRANS, N, NRHS, IA, JA, IB, JB and the array elements IDESCA(1), IDESCA(3),...,IDESCA(8), IDESCB(1) and IDESCB(3),...,IDESCB(8) are all global and so must have the same value on entry to the routine on each processor. The output argument INFO is global and so will have the same value on exit from the routine on each processor. The remaining arguments are local.

3.3 Distribution Strategy

The array A must contain the *LU* factorization of the matrix A_s, previously factorized by F07ARFP (PZGETRF). The *LU* factors must be stored in a cyclic 2-d block distribution (described in the F07 Chapter Introduction), as returned by F07ARFP (PZGETRF). The right-hand sides of the equation, B_s are stored in the array B, in a cyclic 2-d block distribution.

This routine assumes that the data has already been correctly distributed, and if this is not the case will fail to produce correct results. However, the Library provides some utility routines which assist you in distributing data correctly. Descriptions of these routines can be found in Chapters F01 and X04 of the NAG Parallel Library Manual.

4 Arguments

1: TRANS — CHARACTER*1 *Global Input*

On entry: indicates the form of the equations as follows:

if TRANS = 'N', then $A_s X = B_s$ is solved for X;
if TRANS = 'T', then $A_s^T X = B_s$ is solved for X;
if TRANS = 'C', then $A_s^H X = B_s$ is solved for X.

Constraint: TRANS = 'N', 'T' or 'C'.

2: N — INTEGER *Global Input*

On entry: the order of the matrix A_s, n.

Constraint: $0 \leq \mathrm{N} \leq \min(\mathrm{IDESCA}(3), \mathrm{IDESCA}(4), \mathrm{IDESCB}(3))$.

3: NRHS — INTEGER *Global Input*

On entry: the number of right-hand sides, r.

Constraint: $0 \leq \mathrm{NRHS} \leq \mathrm{IDESCB}(4)$.

4: A($*$) — COMPLEX*16 array *Local Input*

Note: the array A is formally defined as a vector. However, you may find it more convenient to consider A as a 2-d array of dimension (IDESCA(9),γ), where
$\gamma \geq \mathrm{numroc}(\mathrm{JA} + \mathrm{N} - 1, \mathrm{IDESCA}(6), p_c, \mathrm{IDESCA}(8), n_p)$. See the Example Program.

On entry: the local part of the *LU* factorization of the matrix A_s as returned by F07ARFP (PZGETRF).

5: IA — INTEGER *Global Input*

On entry: the row index of matrix A, i_A, that identifies the first row of the *LU* factorization of A_s.

Constraints: $1 \leq \mathrm{IA} \leq \mathrm{IDESCA}(3) - \mathrm{N} + 1$ and $\mathrm{mod}(\mathrm{IA} - 1, \mathrm{IDESCA}(5)) = 0$.

6: JA — INTEGER *Global Input*

On entry: the column index of matrix A, j_A, that identifies the first column of the *LU* factorization of A_s.

Constraints: $1 \leq \mathrm{JA} \leq \mathrm{IDESCA}(4) - \mathrm{N} + 1$ and $\mathrm{mod}(\mathrm{JA} - 1, \mathrm{IDESCA}(6)) = 0$.

7: IDESCA(9) — INTEGER array *Local Input*

Distribution: the array elements IDESCA(1) and IDESCA(3),...,IDESCA(8) must be global to the processor grid and the array elements IDESCA(2) and IDESCA(9) are local to each processor.

On entry: the description array for the matrix A. This array must contain details of the distribution of the matrix A and the logical processor grid.

IDESCA(1), the descriptor type. For this routine, which uses a cyclic 2-d block distribution, IDESCA(1) = 1;
IDESCA(2), the BLACS context (ICNTXT) for the processor grid, usually returned by Z01AAFP;
IDESCA(3), the number of rows, m_A, of the matrix A;
IDESCA(4), the number of columns, n_A, of the matrix A;
IDESCA(5), the blocking factor, M_b^A, used to distribute the rows of the matrix A;
IDESCA(6), the blocking factor, N_b^A, used to distribute the columns of the matrix A;
IDESCA(7), the processor row index over which the first row of the matrix A is distributed;
IDESCA(8), the processor column index over which the first column of the matrix A is distributed;
IDESCA(9), the leading dimension of the conceptual 2-d array A.

Constraints:

IDESCA(1) = 1;
IDESCA(3) $\geq$ 0; IDESCA(4) $\geq$ 0;
IDESCA(5) = IDESCA(6); IDESCA(5) $\geq$ 1; IDESCA(6) $\geq$ 1;
$0 \leq$ IDESCA(7) $\leq m_p - 1$; $0 \leq$ IDESCA(8) $\leq n_p - 1$;
IDESCA(9) $\geq$ max(1,numroc(IDESCA(3),IDESCA(5),p_r,IDESCA(7),m_p)).

8: IPIV($*$) — INTEGER array *Local Input*

Note: the dimension of the array IPIV must be at least $\beta +$ IDESCA(5) where,
$\beta =$ numroc(IDESCA(3), IDESCA(5), p_r, IDESCA(7), m_p).

On entry: the pivot indices, as returned by F07ARFP (PZGETRF).

9: B($*$) — COMPLEX*16 array *Local Input/Local Output*

Note: the array B is formally defined as a vector. However, you may find it more convenient to consider B as a 2-d array of dimension (IDESCB(9),γ), where
$\gamma \geq$ numroc(JB+NRHS$-$1,IDESCB(6),p_c,IDESCB(8),n_p). See the Example Program.

On entry: the local part of the right-hand side matrix B which may contain parts of the n by r submatrix B_s.

On exit: the n by r solution matrix X distributed in the same cyclic 2-d block distribution.

10: IB — INTEGER *Global Input*

On entry: the row index of matrix B, i_B, that identifies the first column of the submatrix B_s.

Constraints: $1 \leq$ IB $\leq$ IDESCB(3) $-$ N + 1 and mod(IB $-$ 1, IDESCB(5)) = 0.
The IAth row of the array A and the IBth row of the array B must be located on the same row of the processor grid, i.e.,

$$\text{mod(IDESCA(7) + (IA} - 1)/\text{IDESCA(5)},\ n_p) = \text{mod(IDESCB(7) + (IB} - 1)/\text{IDESCB(5)},\ n_p)$$

11: JB — INTEGER *Global Input*

On entry: the column index of matrix B, j_B, that identifies the first column of the submatrix B_s.

Constraint: $1 \leq$ JB $\leq$ IDESCB(4) $-$ NRHS + 1.

12: IDESCB(9) — INTEGER array *Local Input*

Distribution: the array elements IDESCB(1) and IDESCB(3),...,IDESCB(8) must be global to the processor grid and the array elements IDESCB(2) and IDESCB(9) are local to each processor.

On entry: the description array for the matrix B. This array must contain details of the distribution of the matrix B and the logical processor grid.

IDESCB(1), the descriptor type. For this routine, which uses a cyclic 2-d block distribution, IDESCB(1) = 1;
IDESCB(2), the BLACS context (ICNTXT) for the processor grid, usually returned by Z01AAFP;
IDESCB(3), the number of rows, m_B, of the matrix B;
IDESCB(4), the number of columns, n_B, of the matrix B;
IDESCB(5), the blocking factor, M_b^B, used to distribute the rows of the matrix B;
IDESCB(6), the blocking factor, N_b^B, used to distribute the columns of the matrix B;
IDESCB(7), the processor row index over which the first row of the matrix B is distributed;
IDESCB(8), the processor column index over which the first column of the matrix B is distributed;
IDESCB(9), the leading dimension of the conceptual 2-d array B.

Constraints:

IDESCB(1) = 1;
IDESCB(3) ≥ 0; IDESCB(4) ≥ 0;
IDESCB(2) = IDESCA(2);
IDESCB(5) = IDESCB(6); IDESCB(5) ≥ 1; IDESCB(6) ≥ 1;
$0 \leq$ IDESCB(7) $\leq m_p - 1$; $0 \leq$ IDESCB(8) $\leq n_p - 1$;
IDESCB(8) $\geq$ max(1,numroc(IDESCB(3),IDESCB(5),p_r,IDESCB(7),m_p)).

13: INFO — INTEGER *Global Output*

On exit: INFO = 0 unless the routine detects an error (see Section 5).

5 Errors and Warnings

If INFO $\neq 0$ an explanatory message is output and control returned to the calling program.

INFO < 0

On entry, one of the arguments was invalid:

if the kth argument is a scalar INFO $= -k$;
if the kth argument is an array and its jth element is invalid, INFO $= -(100 \times k + j)$.

This error occurred either because a global argument did not have the same value on all logical processors, or because its value on one or more processors was incorrect.

6 Further Comments

The total number of floating-point operations is approximately $8n^2r$.

6.1 Algorithmic Detail

Forward and backward substitution are used.

6.2 Parallelism Detail

Forward and backward substitution are applied to each block-column of the right-hand sides in parallel.

6.3 Accuracy

For each right-hand side vector b, the computed solution $\hat{x}$ is the exact solution of a perturbed system of equations $(A+E)\hat{x} = b$, where

$$|E| \leq c(n)\epsilon P|L| \cdot |U|,$$

$c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

If x is the true solution, then the computed solution $\hat{x}$ satisfies a forward error bound of the form

$$\frac{\|x-\hat{x}\|_\infty}{\|x\|_\infty} \leq \epsilon c(n)\kappa(A),$$

where $\kappa(A)$ is the condition number of A. See the F07 Chapter Introduction.

7 References

[1] Dongarra J J and Whaley R C (1995) A users' guide to the BLACS v1.0. *LAPACK Working Note 94 (Technical Report CS-95-281)* Department of Computer Science, University of Tennessee, 107 Ayres Hall, Knoxville, TN 37996-1301, USA.
URL: http://www.netlib.org/lapack/lawns/lawn94.ps

[2] Golub G H and Van Loan C F (1989) *Matrix Computations* Johns Hopkins University Press (2nd Edition), Baltimore

8 Example

To solve the system of equations $AX = B$, where

$$A = \begin{pmatrix} -1.34+2.55i & 0.28+3.17i & -6.39-2.20i & 0.72-0.92i \\ -0.17-1.41i & 3.31-0.15i & -0.15+1.34i & 1.29+1.38i \\ -3.29-2.39i & -1.91+4.42i & -0.14-1.35i & 1.72+1.35i \\ 2.41+0.39i & -0.56+1.47i & -0.83-0.69i & -1.96+0.67i \end{pmatrix} \text{ and}$$

$$B = \begin{pmatrix} 26.26+51.78i & 31.32-6.70i \\ 6.43-8.68i & 15.86-1.42i \\ -5.75+25.31i & -2.15+30.19i \\ 1.16+2.57i & -2.56+7.55i \end{pmatrix}.$$

A is first factorized by F07ARFP.

The example uses a 2 by 2 logical processor grid and a block size of 2 for both A and B.

Note: the listing of the Example Program presented below does not give a full pathname for the data file being opened, but in general the user must give the full pathname in this and any other OPEN statement.

8.1 Example Text

```
*     F07ASFP Example Program Text
*     NAG Parallel Library Release 2. NAG Copyright 1996.
*     .. Parameters ..
      INTEGER           NIN, NOUT
      PARAMETER         (NIN=5,NOUT=6)
      INTEGER           DT
      PARAMETER         (DT=1)
      INTEGER           MB, NB
      PARAMETER         (MB=2,NB=MB)
      INTEGER           NMAX, IAROW, IACOL, LDA, LDB, NRHMAX, LW
      PARAMETER         (NMAX=8,IAROW=0,IACOL=0,LDA=NMAX,LDB=NMAX,
     +                  NRHMAX=2,LW=NMAX)
*     .. Local Scalars ..
      INTEGER           IA, IB, ICNTXT, IFAIL, INFO, JA, JB, MP, N, NP,
```

```
     +                 NRHS
      LOGICAL          ROOT
      CHARACTER        TRANS
      CHARACTER*80     FORMAT
*     .. Local Arrays ..
      COMPLEX*16       A(LDA,NMAX), B(LDB,NRHMAX), WORK(LW)
      INTEGER          IDESCA(9), IDESCB(9), IPIV(NMAX+MB)
*     .. External Functions ..
      LOGICAL          Z01ACFP
      EXTERNAL         Z01ACFP
*     .. External Subroutines ..
      EXTERNAL         F07ARFP, F07ASFP, X04BRFP, X04BSFP, Z01AAFP,
     +                 Z01ABFP
*     .. Executable Statements ..
      ROOT = Z01ACFP()
      IF (ROOT) THEN
         WRITE (NOUT,*) 'F07ASFP Example Program Results'
         WRITE (NOUT,*)
      END IF
*
      MP = 2
      NP = 2
      IFAIL = 0
*
      CALL Z01AAFP(ICNTXT,MP,NP,IFAIL)
*
      OPEN (NIN,FILE='f07asfpe.d')
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, NRHS, FORMAT

      IF (N.LE.NMAX) THEN
*
*        Set the array descriptor of A
*
         IDESCA(1) = DT
         IDESCA(2) = ICNTXT
         IDESCA(3) = N
         IDESCA(4) = N
         IDESCA(5) = MB
         IDESCA(6) = NB
         IDESCA(7) = IAROW
         IDESCA(8) = IACOL
         IDESCA(9) = LDA
         IA = 1
         JA = 1
*
*        Read A from the data file
*
         IFAIL = 0
         CALL X04BRFP(NIN,N,N,A,1,1,IDESCA,IFAIL)
*
*        Factorize the matrix
*
         CALL F07ARFP(N,N,A,IA,JA,IDESCA,IPIV,INFO)
*
         IF (INFO.EQ.0) THEN
            TRANS = 'N'
```

```
*
*           Set the array descriptor of B
*
            IDESCB(1) = DT
            IDESCB(2) = IDESCA(2)
            IDESCB(3) = N
            IDESCB(4) = NRHS
            IDESCB(5) = MB
            IDESCB(6) = NB
            IDESCB(7) = IAROW
            IDESCB(8) = IACOL
            IDESCB(9) = LDB
            IB = 1
            JB = 1
*
*           Read B from data file
*
            IFAIL = 0
            CALL X04BRFP(NIN,N,NRHS,B,1,1,IDESCB,IFAIL)
*
            CALL F07ASFP(TRANS,N,NRHS,A,IA,JA,IDESCA,IPIV,B,IB,JB,
     +                   IDESCB,INFO)
            IF (INFO.EQ.0) THEN
*
*               Print solution(s)
*
               IF (ROOT) THEN
                  WRITE (NOUT,*) 'Solution(s)'
                  WRITE (NOUT,*)
               END IF
               IFAIL = 0
*
               CALL X04BSFP(NOUT,N,NRHS,B,IB,JB,IDESCB,FORMAT,WORK,
     +                      IFAIL)
*
            ELSE
               IF (ROOT) WRITE (NOUT,*)
     +             'Unable to solve triangular system'
            END IF
         ELSE
            IF (ROOT) WRITE (NOUT,*) 'Matrix is singular'
         END IF
*
      END IF
*
      CLOSE (NIN)
*
      CALL Z01ABFP(ICNTXT,'N',IFAIL)
*
      STOP
      END
```

8.2 Example Data

```
F07ASFP Example Program Data
 4  2 'F7.4'                                    :Values of N, NRHS and FORMAT
 (-1.34, 2.55) ( 0.28, 3.17) (-6.39,-2.20) ( 0.72,-0.92)
 (-0.17,-1.41) ( 3.31,-0.15) (-0.15, 1.34) ( 1.29, 1.38)
```

```
(-3.29,-2.39) (-1.91, 4.42) (-0.14,-1.35) ( 1.72, 1.35)
( 2.41, 0.39) (-0.56, 1.47) (-0.83,-0.69) (-1.96, 0.67)  :End of matrix A
(26.26, 51.78) (31.32, -6.70)
( 6.43, -8.68) (15.86, -1.42)
(-5.75, 25.31) (-2.15, 30.19)
( 1.16,  2.57) (-2.56,  7.55)                             :End of matrix B
```

8.3 Example Results

```
F07ASFP Example Program Results

Solution(s)

( 1.0000, 1.0000) (-1.0000,-2.0000)
( 2.0000,-3.0000) ( 5.0000, 1.0000)
(-4.0000,-5.0000) (-3.0000, 4.0000)
( 0.0000, 6.0000) ( 2.0000,-3.0000)
```